
9 Map and Grep

9.1 Map

map is a function that makes it easy to copy all of the elements from one array to another and transform each element during the copy procedure.  For example, say you have an array of ten numbers and you want a copy of this array with each number multiplied by 2.  If you used a for or foreach loop, this is how you would do it:

my @a = (1..10);

my @b;

for( my $i=0; $i<@a; $i++ ) {


$b[$i] = 2 * $a[$i];

}

my @a = (1..10);

my @b;

foreach ( @a ) {


push @b, 2 * $_;

}

Using map, you would do it this way:

my @a = (1..10);

my @b = map { $_ * 2 } @a;

map has two "arguments", a code block specifying the operation that is done on each array element, and an array or list, of element map is to manipulate.  I put "arguments" in quotation because these are subroutine arguments in the normal sense, notice that there is no comma separating the code block and array.  The return value from map is a list of the modified elements of the first array.

In the code block, each array element is assigned to the implicit variable $_.  The result from any evaluations is sent to the output array.  In the above example,  the result of $_ * 2 is 2, 4, 6, etc.  We don't need to assign this expression to a variable for it to be stored.

You are allowed to have multiple statements in the code block.  The result of the last statement is what will be sent to the output array.  If you code block contains multiple statements, the best coding practice is to store $_ to a named variable and use that variable for the remaining statements.

#

# @b will contain 4 6 8 10 12 14 16 18 20 22

#

my @a = (1..10);

my @b = map { my $n = $_*2; $n+2 } @a;
You must be careful when using $_.  It is aliased to the original elements in the input array.  Any changes you make to $_ will change the original elements of the input array.

#

# Both @a and @b will contain 2 4 6 8 10 12 14 16 18 20

#
my @a = (1..10);

my @b = map { $_*=2 } @a;
If your intent was to specifically modify the elements of @a, and you didn't care about @b, you could take advantage of this property.

#

# Square all of the values in @a

#

map { $_ *= $_ } @a;
While the above syntax is perfectly legal, some Perl experts consider it to be a bad idea to modify $_ in the middle of a transform
.  The following syntax is more often seen when transforming an array:

#

# Square all of the values in @a

#
@a = map { $_ * $_ } @a;
Because the input array elements are automatically assigned to $_, you will often see map transformations that are very terse if the operators in the code block know how to handle $_.

#

# Make all of the names uppercase and store them in @b

#

my @a = qw(jon jane jun jack jill);

my @b = map { uc } @a;
9.2 Map and regular expressions

When you combine the memory feature of regular expressions and map, you can do many useful things in one line of code.

#

# @files contains a list of file names.

# Find all of the files in @files with a *.txt

# extension and store the base names in @names

#

@names = map { /(.*)\.txt/ } @files;
Every element in  @files is stored in $_.  The regular expression operates on $_ and stores the base name of any *.txt file in $1.  $1 is then assigned to @names as one array element.

If @files was a 100 element array but only contained 5 *.txt file names, @names would only be 5 elements long, not 100 with 95 empty array elements.  If @files didn't contain any *.txt files, @names would be a zero length array.

If you have multiple memory parentheses in your regular expression, each memory variable $1, $2, $3, etc. will all be stored in the output array.

#

# @files contains a list of file names.

# Find all of the files in @files with any extension

# and store the base names and file extensions in @names.

#

# In this examples, @names will contain

# (a1, txt, b2, doc, c3, xls, d4, pdf)

#

my @files = qw(a1.txt b2.doc c3.xls d4.pdf);

my @names = map { /(.*)\.(.*)/ } @files;
Since we are talking about file names, we should note that the map will operate on anything that can be used in a list context.  For example, <*> will return a list of files in the current directory.  We can use this to store the base names of these files in an array:

my @names = map { /(.*)\..*/ } <*>;

9.3 Map and hashes

Since hashes can be initialized with a list or array, you can use map to assign key-value pairs to a hash.

#

# This code will take the Linux passwd file from

# /etc/passwd, and extract each accounts username

# and real name.  These two names are stored in

# a hash where the usernames are the keys and the

# real names are the values.

#

# A line of passwd typically looks like this:

#

# curleejl:x:1028:5526:Joanna Curlee:/home/curleejl:/bin/bash
#

# Each line is colon delimited.  The first item is the 

# username.  The fifth is the real name.

#

# After the transform, one hash element will look like this

#

# $h{'curleejl'} = 'Joanna Curlee'

#

my @pswd = `cat /etc/passwd`;

my %h = map {/^([^:]+):[^:]+:[^:]+:[^:]+:([^:]+):/} @pswd;

In the previous code example I used a regular expression to store the username and real name in $1, $2.  This code is needlessly complex and error prone.  If I rewrite the code in the following manner, it is more obvious what I'm doing and I'm more likely to write bug free code the first time I write it:

my @pswd = `cat /etc/passwd`;

my %h = map {my @a = split(':',$_); $a[0] => $a[4]} @pswd;

In this example, we split each line of passwd on the colon delimiter and pick out $a[0] and $a[4], the username and real name.  The second statement in the map code block, $a[0] => $a[4] is assigned to %h.  Because of the interchangeability of arrays and hashes, we could have put '$a[0], $a[4]' (without quotes) as the last statement too.  I prefer using the => syntax since we are assigning the data to a hash.

9.4 Grep

grep has a similar interface to map but it's intended use is entirely different.  While map is designed to transform all of the elements of an array and copy the modified elements to a new array, grep is designed to copy only those elements that meet a specified condition.  Another way of saying this is that grep is an array filter.

#

# Only copy those elements of @a to @b that start

# with a '#'

#

@b = grep { /^#/ } @a;

As you might have inferred from this example, the elements of $a are aliased to $_, just like map (and the same warnings apply, too).

You can't use grep to transform the elements that are copied like you can with map.  This code won't double the numbers in @a and assign them to @b:

#

# This code will give you error messages and 

# cause havoc on the contents of @a

#

@a = qw(one 1 two 2 three 3);

@b = grep { /\d/; $_ *= 2 } @a;
If you wanted to double the elements of @a with digits, you can combine grep and map into one statement.

#

# @b will contain 2 4 6

#
@a = qw(one 1 two 2 three 3);

@b = map { $_ *= 2 } grep { /\d/ } @a;
� Although I haven't seen any specific examples, I can image that there maybe some cases where complicated transformations of $_ may not go as expected.

